14.Reducing code size(在所有的處理器中)





	在cache那一章裡面說過，程式碼的cache是8或是16 kb。如果你沒有辦法把你程式碼的critical part都放在程式碼的cache裡，你就得考慮縮減你的程式碼了。





	因為資料常數在32位元程式碼中佔4 bytes的空間，而在16位元程式碼中只佔2 bytes的空間，所以32位元的程式碼通常都比16位元的大。然而16位元的程式碼也有其他的問題，像是前加字(prefixes)以及同時處理相鄰兩個位元的問題(在imperfect pairing那一章中有介紹)。底下會介紹一些縮減程式碼的方法。





(一) Sign-extanded byte 的重要





	Sign-extanded byte 就是以 8 bits 二進位數字的第一個位元來表示正負的數，所以其值在 -128 到 +127 之間。





	跳躍位置(jump address)、資料位置(data address)、以及資料常數(data constants)的值如果在-128到+127之間(也就是可以以第一個bit來表示正負的數)，就可以使用比較小的空間來儲存。





	短距離的跳躍在程式碼中佔2 bytes，但是超過127 bytes的跳躍就會佔5或6 bytes的空間。





	同樣的，如果一個資料位置是在-128到+127之間，就可以佔用比較小的空間。例如將：





MOV EBX,DS:[100000] 


ADD EBX,DS:[100004] 	  ; 12 bytes





減化為：





MOV EAX,100000


MOV EBX,[EAX]


ADD EBX,[EAX+4] 		  ; 10 bytes





	如果你使用靜態記憶體以及絕對位置，把資料排序在堆疊裡，並且用EBP和ESP當指位器也會讓你的程式碼小一些，當然，你的資料要在指位器的前後127 bytes內。使用PUSH和POP來讀寫暫存資料則會有更多的好處。





	資料常數如果在-128到+127之間也會佔用較少的空間。大多數有即時運算元(immediate operands)的指令，如果運算元的大小在-128到+127之間，所佔的空間也會變小。例如：





PUSH 200         ; 5 bytes


PUSH 100         ; 2 bytes





ADD EBX,128     ; 6 bytes


SUB EBX,-128     ; 3 bytes





但是MOV指令沒有辦法用這種方式縮減。例如：





MOV EAX, 1              ; 5 bytes





如果想要縮減空間，可以把程式改成：





XOR EAX,EAX 


INC EAX   		; 3 bytes


或是


PUSH 1


POP EAX        ; 3 bytes





如果之前就已經知道暫存器的值，就可以用一個運算指令替代一個跟著4 bytes即時運算元的MOV指令。例如：





MOV     [mem1],0        ; 10 bytes


        MOV     [mem2],-1       ; 10 bytes


        MOV     EAX,100         ;  5 bytes


        MOV     EBX,150         ;  5 bytes


可以改成：


        XOR     EAX,EAX         ;  2 bytes


        MOV     [mem1],EAX      ;  5 bytes


        DEC     EAX             ;  1 byte


        MOV     [mem2],EAX      ;  5 bytes


        ADD     EAX,101         ;  3 bytes


        LEA     EBX,[EAX+50]    ;  3 bytes





如果一個常數被使用一次以上，你當然應該把它載入一個暫存器。例如：





MOV	 DWORD PTR [EBX],0


MOV	 DWORD PTR [EBX+4],0    ;13 bytes





XOR 	EAX,EAX 


MOV 	[EBX],EAX 


MOV 	[EBX+4],EAX     		;7 bytes





(二) 只佔 1 byte 的指令





明顯的，不同的指令會有不同的長度。


下面的指令都只有1 byte的長度，可以好好利用：





		PUSH reg, POP reg, INC reg32, DEC reg32





使用8 bits暫存器的INC和DEC指令佔兩個bytes，所以INC EAX會比INC AL短。





XCHG EAX, reg也是一個1 byte的指令，所以使用的空間比MOV EAX, reg少，但是執行的速度比較慢。





(三) 使用的暫存器不同也會影響指令大小





有些指令在使用累加暫存器(accumulator)的時候會比使用其它暫存器少一個byte。例如：





MOV EAX,DS:[100000]  比  MOV EBX,DS:[100000]  小


ADD EAX,1000         比  ADD EBX,1000  小





有些指令在只使用一個基底指位器的時候會比


使用scaled index register的時候、


同時使用基底指位器和index暫存器、


(3)或是用ESP當做基底指位器時少用一個byte及位移(displacement)。例如：





MOV EAX,[array][EBX]  比  MOV EAX,[array][EBX*4]  小


MOV EAX,[EBP+12]      比  MOV EAX,[ESP+12]  小





使用ESP做為基底指位器，並且沒有位移也沒有index會比其它暫存器多用一個byte:





MOV EAX,[EBX]    比  MOV EAX,[EBP] 小，但是  


MOV EAX,[EBX+4]  和  MOV EAX,[EBP+4]. 則一樣大








